
Abstract

Tabling is a commonly used technique in logic programming for avoiding cyclic behavior of

logic programs and enabling more declarative program de�nitions. Furthermore, tabling often

improves computational performance. Rational term are terms with one or more in�nite sub-

terms but with a �nite representation. Rational terms can be generated in Prolog by omitting

the occurs check when unifying two terms. Applications of rational terms include de�nite clause

grammars, constraint handling systems, and coinduction. In this paper, we report our extension

of YAP's Prolog tabling mechanism to support rational terms. We describe the internal repre-

sentation of rational terms within the table space and prove its correctness. We then use this

extension to implement a tabling based approach to coinduction. We compare our approach with

current coinductive transformations and describe the implementation. In addition, we present

an algorithm that ensures a canonical representation for rational terms.

KEYWORDS: Tabling, Rational Terms, Coinduction, Implementation.

1 Introduction

Tabling (Chen and Warren 1996) is a recognized and powerful implementation tech-

nique that solves some limitations of Prolog's operational semantics in dealing with

left-recursion and redundant sub-computations. Tabling based models are able to re-

duce the search space, avoid looping, and always terminate for programs satisfying the

bounded term-size property1. Tabling consists of saving and reusing the results of sub-

computations during the execution of a program. This is accomplished by storing the

calls and the answers to tabled subgoals in a proper data structure called the table space.

From as early as (Colmerauer 1982; Ja�ar and Stuckey 1986), Prolog implementers

have chosen to omit the occurs check in uni�cation. This has resulted in generating cyclic

terms known as rational terms or rational trees in Prolog. While the introduction of cyclic

terms in Prolog was unintentional, soon after applications for cyclic terms emerged in

�elds such as de�nite clause grammars (Colmerauer 1982; Giannesini and Cohen 1984),

1 A logic program has the bounded term-size property if there is a function f : N → N such that
whenever a query goal Q has no argument whose term size exceeds n, then no term in the derivation
of Q has size greater than f(n).

constraint programming (Meister and Frühwirth 2006; Bagnara et al. 2001) and coin-

duction (Gupta et al. 2007). But support for rational terms across Prolog systems varies

and often fails to provide the functionality required by most applications. Two common

problems are the lack of support for printing query bindings with rational terms and the

lack of a tabling mechanism that supports rational terms (Moura 2013). Furthermore,

several Prolog features are not designed for compatibility with rational terms and can

make programming using rational terms challenging and cumbersome.

In this paper, we �rst present an extension to the tabling mechanism of YAP Pro-

log (Santos Costa et al. 2012) to support rational terms. To the best of our knowledge,

this is the �rst Prolog built-in tabling system that supports rational terms. The tabling

system of XSB (Swift and Warren 2012) handles in�nite terms by de�ning a limit on the

term size stored within the table space. While this approach allows tabling to work with

goals containing rational terms, it does not store a rational term within the table space

and, as a result, the returned answers are not the expected ones. Our approach extents

the table space to represent rational terms internally. Consequently, answers containing

rational terms are properly stored and returned by the tabling mechanism.

Furthermore, by having extended tabling to support rational terms, we present a novel

approach for calculating the greatest �xed point of goals by using the internal tabling

mechanism. In this way, we propose a novel and e�cient transformation to implement

coinduction. Previous research presented di�erent solutions to support coinduction (Si-

mon et al. 2007; Gupta et al. 2007; Ancona 2013; Moura 2013) but, to the best of our

knowledge, our approach is the �rst one that uses a native transformation, allowing a

more elegant and e�cient implementation. In addition, we present an algorithm that

ensures a canonical representation for rational terms.

The remainder of the paper is organized as follows. First in Section 2, we introduce

some background concepts about tabling, rational terms and coinduction. Next, Section 3

describes our extension to support rational terms within the tries of the table space.

Then, we introduce our proposal to implement coinduction through the internal tabling

mechanism in Section 4. Section 5 addresses the issue of rational terms in canonical form

and �nally, Section 6, concludes by discussing related and future work.

2 Background

In this section, we present some background information on tabling, rational terms, and

coinduction.

2.1 Tabling and Tries

The basic idea behind tabling is straightforward: programs are evaluated by storing

answers for tabled subgoals in an appropriate data space called the table space. Repeated

calls to tabled subgoals are resolved whenever possible by consuming the answers already

stored in the table space instead of being re-evaluated against the program clauses. During

this process, as new answers are found, they are stored in their tables and later returned

to all repeated calls.

The design of the table space data structures is critical for achieving an e�cient tabling

implementation. In YAP, as in most tabling systems, the table space representation is

based on tries (Ramakrishnan et al. 1999). A trie is a tree structure where each di�erent

path through the tree nodes corresponds to a term described by the tokens labeling

the nodes traversed. For example, the tokenized form of the term path(X,1,f(Y)) is the

sequence of 5 tokens path/3, VAR0, 1, f/1 and VAR1, where each variable is represented as a

distinct VARi constant. Two terms with common pre�xes will branch o� from each other

at the �rst distinguishing token. Consider, for example, a second term path(Z,1,b). Since

the main functor and the �rst two arguments, tokens path/3, VAR0 and 1, are common to

both terms, only one additional node will be required to fully represent this second term

in the trie, thus saving the space that would be taken by three nodes in this case.

YAP implements tables using two levels of tries (Ramakrishnan et al. 1999). The �rst

level, the subgoal trie, stores the tabled subgoal calls. The second level, the answer trie,

stores the answers for a given call. More speci�cally, each tabled predicate has a table

entry data structure assigned to it, acting as the entry point for the predicate's subgoal

trie. Each di�erent subgoal call is then represented as a unique path in the subgoal

trie, starting at the table entry and ending in a subgoal frame data structure, with the

argument terms being stored within the path's nodes. The subgoal frame data structure

acts as an entry point to the answer trie. Contrary to subgoal tries, answer trie paths

hold just the substitution terms for the free variables that exist in the argument terms

of the corresponding call.

f/1

VAR0

VAR0

1

VAR1

subgoal
trie

subgoal frame for
p(VAR0,1,VAR1)

a

b

answer
trie

1st
argument

2nd
argument

3rd
argument

substitution
term for

1st argument

substitution
term for

3rd argument

table entry for
p/3

Fig. 1. YAP's table space

organization

An example for a tabled predicate p/3 is shown in

Fig. 1. Initially, the table entry for p/3 points to an empty

subgoal trie. When the subgoal p(X,1,Y) is called, three

trie nodes are inserted to represent the arguments in the

call: one for the variable X (VAR0), a second for the in-

teger 1, and third one for the variable Y (VAR1). Since

the predicate's functor term is already represented by its

table entry, we can avoid inserting an explicit node for

p/3 in the subgoal trie. Finally, the leaf node is set to

point to a subgoal frame, from where the answers for the

call will be stored. The example shows two answers for

p(X,1,Y): {X=a, Y=f(VAR0)} and {X=a, Y=b}. Since both

answers have the same substitution term for argument

X, they share the top node in the answer trie. For ar-

gument Y, each answer has a di�erent substitution term

and, thus, a di�erent path is used to represent each.

2.2 Rational Terms

Rational terms, also known as rational trees or cyclic terms, are in�nite terms that can

be �nitely represented. They can include any �nite sub-term but have at least one in�nite

sub-term. Rational terms in logic programming appeared as a side e�ect of omitting the

occurs check in uni�cation. By omitting the occurs check, uni�cation can return cyclic

terms. A simple example is L=[1|L], where the variable L is instantiated to an in�nite list

of ones. Prolog implementers omitted the occurs check in order to reduce the uni�cation

complexity from O(SizeTerm1 +SizeTerm2) to O(min(SizeTerm1, SizeTerm2)), but since

then these cyclic terms found application in areas such as coinduction, natural language

processing or �nite-state automata, among others.

Several Prolog systems support the creation and uni�cation of rational terms. These

systems also provide built-in predicate implementations capable of handling rational

terms without falling into in�nite computations. Still, several common predicates such

as member/2 cannot handle rational terms.

member(E, [E|_]).
member(E, [_|T]) :- member(E, T).

Consider the call L=[1,2|L], member(E, L). The second clause of member/2 spawns a

new member call for every next su�x of the in�nite list L, generating an in�nite number

of calls and answers for member/2. In order to handle the occurring cycles, one could

implement the following member/2 predicate.

member(E, L) :- member(E, L, []).

member(E, [E|_], _).
member(_, L, S) :- in_stack(L, S), !, fail.
member(E, [H|T], S) :- member(E, T, [[H|T]|S]).

With in_stack/2 predicate being de�ned as:

in_stack(E, [H|_]) :- E == H.
in_stack(E, [_|T]) :- in_stack(E, T).

The second clause of member/3 checks whether the input list of the member call is

repeated. If so, it terminates the execution by removing any left choice points and then

fails. Using our example list L=[1,2|L], it will permit only the two unique su�xes of the

list L. Unfortunately, this is more of an ad-hoc solution instead of a clean scalable solution

that deals with generic rational terms.

2.3 Coinduction

Recently, we have seen an increase of interest in coinduction. Several new systematic

approaches have appeared and important applications have been noted. co-SLD extends

logic programming to allow reasoning over in�nite and cyclic structures. For this paper,

we follow similar semantics like the ones proposed by (Gupta et al. 2007) for co-SLD: the

coinductive hypothesis rule states that if while proving a goal G (ancestor goal) a subgoal

G' (current goal) is found, such as goal G is a variant of G', then the subgoal G' has

coinductive success.

We have seen several approaches to implement co-SLD with the use of program trans-

formation (Simon et al. 2007; Gupta et al. 2007; Moura 2013) or by using a meta caller

approach (Ancona 2013). While SLD resolution proves a query and returns the answers of

the least �xed points, co-SLD resolution succeeds and returns the answers of the greatest

�xed points (Simon et al. 2007). As an example, consider the following logic program

and the query bin(X).

bin([]).
bin([0|T]) :- bin(T).
bin([1|T]) :- bin(T).

SLD resolution recognizes all �nite lists containing only 0s and 1s. Now, if we remove

the base case, the program does not have a least �xed point and, as a consequence,

SLD resolution will not be able to solve it. As pointed out by (Gupta et al. 2007), there

are two reasons why the program is not solved. First, because the Herbrand universe

does not allow in�nite terms and, second, because the Herbrand universe does not allow

in�nite proofs. co-SLD allows in�nite terms and in�nite proofs, as a result the query

bin(X) succeeds and resolution returns two answers that represent the minimal greatest

�xed points, {X=[0|X], X=[1|X]}. However, any in�nite pattern of 0s and 1s that can

be expressed as a rational term is recognized by the bin/1 coinductive predicate, as

exempli�ed below.

% generating solutions for the coinductive bin/1 predicate:
Query: ?- bin(X).
Answers: X = [0|X] ? ;

X = [1|X].

% recognizing solutions for the coinductive bin/1 predicate:
Query: ?- X = [0,1,0,1,0,0,0|X], bin(X).
Answer: X = [0,1,0,1,0,0,0|X].

3 Extending Tabling to Support Rational Terms

To the best of our knowledge, no current tabling system supports rational terms. A

tabled subgoal call or tabled answer containing rational terms would result in an in�nite

computation. The reason for this behavior is that the table space data structures were

not designed to support the �nite storing of in�nite terms. YAP is thus the �rst tabling

system that handles rational terms by extending the table space to support the internal

representation of in�nite terms.

Consider, for example, the de�nition given in (Gupta et al. 2007) for a comember/2

coinductive predicate:

% comember(Element, List) is true iff Element appears infinite times in List
:- coinductive(comember/2).
comember(H, L) :- drop(H, L, L1), comember(H, L1).

% drop(Element, List, Rest) is true iff Element can be dropped from List resulting
% in the list Rest
:- table(drop/3).
drop(H, [H|T], T).
drop(H, [_|T], T1) :- drop(H, T, T1).

Without tabling with support for rational terms, the auxiliary predicate drop/3 can

not be used by the comember/2 predicate to enumerate all the distinct elements of a

(rational) list. Only the �rst element would be enumerated. To avoid tabling, one could

write drop/3 as follows, by using the in_stack/2 auxiliary predicate, but certainly writing

such a procedural predicate is not in the spirit of logic programming. Similarly, the

solution presented by (Ancona 2013) that avoids tabling by using coinductive success

hook predicates requires from the programmer a good understanding of the coinductive

meta-interpreter or coinductive transformation mechanics. Ancona's coinductive success

hook predicates are not illustrated here for brevity. The crucial concept behind them, is

to enable the programmer to modify the behavior when a cycle is detected.

drop(E, L, NL) :- drop(E, L, NL, []).
drop(_, L, _, S) :- in_stack(L, S), !, fail.
drop(E, [E|T], T, _).
drop(E, [H|T], T1, S) :- drop(E, T, T1, [[H|T]|S]).

Motivated by the above example and the goal to elegantly and e�ciently support

coinduction for YAP at the low level engine, our �rst step was to extend YAP's internal

tries representation to support rational terms. So far, tries do not support cyclic terms

and inserting one such term in a trie results in trapping the program in a cycle. In order

for tries to support cyclic terms, we �rst de�ned a representation of cyclic tokens within

the tries; we then extended the trie check/insert algorithm to detect the tokens where

a cycle occurs and at which upper trie token it should point to; last, we adjusted the

mechanism that allows to reconstruct a term from the stored tokens within the trie to

also handle cyclic terms.

head term

tail term

pair term

1

pair term

2

pair term

3

var term

entry point
for L PAIR

1

PAIR

trie entry

2

PAIR

3

RT_PTR

(b)(a)

Heap

Fig. 2. YAP's representation for the cyclic

term L=[1,2,3|L] in the (a) heap and (b)
the trie

To better explain our trie representation

for cyclic terms let's have a look at the heap

representation. At the low level engine, a cy-

cle in YAP appears as a variable term cell

uni�ed with the term that is repeated. Fig-

ure 2(a) shows YAP's internal representa-

tion for the cyclic term L=[1,2,3|L] as it is

stored in the heap.

Current trie de�nitions do not store

pointer tokens by default. We extend the

trie de�nition so that it can store a pointer

token that points to a trie node. As no other

trie nodes are storing pointers, we can safely

assume that such a pointer token represents

a rational term cycle. In order to represent

the cycle within the trie, we store in the

node of the cyclic token a pointer to the

trie node of the pointed token. Figure 2(b) illustrates the cyclic term L=[1,2,3|L] as it is

stored in the trie representation2 with RT_PTR illustrating the rational term pointer token.

In order for our approach to be safe we consider Proposition 1.

Proposition 1

Cycles in a Prolog term can only point at a list (pair) term or a functor term.

2 For the sake of simplicity, we describe YAP's original trie implementation for lists. In fact, our current
implementation extends the optimized compact lists representation for tries as described in (Raimundo
and Rocha 2010).

Proof

If a variable term cell points to an atomic term or to an unbound variable then it would

be dereferenced to that atomic term or unbound variable.

We detect the cycles upon the insertion of a term in the trie. To do so, we �rst memorize

the heap address of every list or functor term that we encounter tuppled with the address

of the trie node that represents the list or functor token in the trie. Before inserting a

new list or functor token in the trie, we check whether we have already inserted a node

for it. If so, we return the address of the corresponding trie node and create a new trie

node containing a pointer token with the address of the returned trie node.

Besides storing a cyclic term, the tabling system must also be able to return the

same cyclic term when asked. In YAP Prolog, to reconstruct a term from the trie, the

tabling mechanism reads the trie branch bottom up and constructs the subterms stored

in each token incrementally directly in the Prolog heap. The original term reconstruction

mechanism assumes that no reference pointers are stored within the trie nodes. As the

construction of the term is bottom up our system does not know the reference of a

cycle before fully constructing the term. In order to tackle this issue when a reference

pointer is found, we create a new unbound variable in the heap and memorize the heap

address and the trie node that has the token which will be the entry point of the term

that this variable must be uni�ed. When the term has been fully constructed, we unify

that unbound variable with the term generated from the corresponding trie node that

the pointer token is pointing at. In order for our approach to be safe we considered

Proposition 2.

Proposition 2

When traversing a Prolog term from the heap entry point, cyclic references point at

preceding heap locations.

Proof

Indeed, if while traversing a Prolog term, starting from the entry point, a variable term

cell points to a following heap location it does not introduce a cycle.

Returning to our member/2 example from Section 2.2, we now present an alternative

more elegant solution to handle lists with rational terms.

:- table(member/2).
member(E, [E|_]).
member(E, [_|T]) :- member(E, T).

Please note that tabling not only remembers answers but also handles introduced

cycles. Similarly, any predicate that needs to process a list which contains rational terms

can use tabling to avoid using a similar mechanism like the one presented at Section 2.2.

4 A Tabling Based Approach to Coinduction

In this section, we describe a modi�ed tabling strategy that instead of computing the

least �xed point of a logic program, computes the greatest �xed point. In order to com-

pute the greatest �xed point of a logic program, it is customary to either de�ne a program

transformation (Simon et al. 2007; Gupta et al. 2007; Moura 2013) or implement a meta-

interpreter (Ancona 2013) that solves the logic program. Both approaches have been

presented in the past and, to the best of our knowledge, there is no previous implemen-

tation that computes the greatest �x point within the WAM.

In order to better describe our tabling based approach, we �rst present in Alg. 1 the

program transformation described in (Moura 2013). There are three key features in the

transformation: (i) it uses a stack to keep track the ancestor calls (also called coinductive

hypothesis) of the goal that the greatest �x point is under computation; (ii) the stored

ancestor calls in the stack are then used for detecting cycles (for coinduction purposes a

cycle occurs when an ancestor call uni�es with the current call) in the execution of the

goal; and (iii) when a cycle is detected, it is handled as a success and uni�es the called

goal with the repeated ancestor goal.

Input: a coinductive logic predicate p/1

:- coinductive(p/1).
p(Args) :-

% the body/2 predicate abstracts the processing of the predicate arguments
body(Args, NewArgs),
p(NewArgs).

Output: p/1 transformed to call an auxiliary predicate that implements coinduction

p(Args) :-
% start with an empty stack of coinductive hypothesis
p(Args, []).

p(Args, Stack) :-
body(Args, NewArgs),
p_coinduction_perflight(NewArgs, Stack).

p_coinduction_perflight(Args, Stack) :-
( member(p(Args), Stack) *->

% coinductive success;
% the (*->)/2 soft-cut construct supports finding of multiple solutions
true

; % add current goal to the stack of coinductive hypothesis and continue
p(Args, [p(Args)|Stack])

).

Alg. 1: The coinductive program transformation of (Moura 2013).

Algorithm 1 is akin both to the cycle handling used in the PTTP (Stickel 1988) and

in tabling systems. Tabling memoizes subgoals in a table and checks whether there is

a repeated call in order to break the cycle, thus allowing cyclic programs to terminate.

Conceptually, the subgoal memoization mechanism of tabling used to perform cycle de-

tection is similar to the collection of calls in the stack used by Alg. 1. One slight di�erence

is that Alg. 1 is using uni�cation to detect repeating goals and YAP's tabling mechanism

uses variant checking3, which may alter the depth of derivations needed to detect the

coinductive cycle. To compute the greatest �xed point of a logic program by reusing the

mechanism of tabling, we thus noticed that we need to do two main modi�cations to

3 Two goals are said to be variants if they are the same up to variable renaming.



Tabling, Rational Terms, and Coinduction Finally Together! 9

YAP's tabling mechanism. First, we need to modify the behavior of the cycle detection

mechanism in order to succeed the subgoal when a cycle was detected and, second, we

need to unify the arguments of the original call of the goal with the arguments of the

repeated subgoal where the cycle was detected. While the �rst modi�cation is easy to

motivate, the second modi�cation is not so obvious. To clarify the function and necessity

of the second modi�cation, we use the example shown in Fig. 3.

:- coinductive(bin/1).
bin([0|T]) :- bin(T).
bin([1|T]) :- bin(T).

Query: ?- bin(L).
Answers: L = [0|L] ? ;

L = [1|L].
Query: ?- X=[0,1,0,0|X], bin(X).
Answer: X = [0,1,0,0|X].

?- bin(L).

1. bin(T)

2. L = [0|L]

3. bin(T)

4. L = [1|L]

L = [0|T] L = [1|T]

L = T L = T

Fig. 3. The coinductive predicate bin/1 and its coinductive SLG tree

In this example, the uni�cation of the query goal bin(L) with the head of both clauses

for bin/1 leaves the tail T of the list unbound (L=[0|T] and L=[1|T] for the �rst and second

clause, respectively). The cycle is then detected when the subgoal bin(T) is called next

(steps 1 and 3 in Fig. 3). Unifying the arguments for the initial query goal bin(L) with

the subgoal bin(T) leads to unifying L with T, which results in the answers L=[0|L] and

L=[1|L] for the query goal (steps 2 and 4 in Fig. 3).

generator
choice point

N

subs_var_N

var cell

var cell

Heap

subs_var_1

...

N

subs_var_N

subs_var_1

...

...

Choice point
stack

var cell

var cell

coinductive
unification

substitution
factor for

repeated call

substitution
factor for
query goal

Fig. 4. Using the substitution factors to

implement coinductive uni�cation

Now that we have identi�ed the

necessary behavior for computing the

greatest �xed point with co-SLG, we

present how it is actually implemented.

As explained in section 2.1, the tabling

mechanism retains a state called sub-

goal frame for each subgoal call that

is proving. In particular, this includes

the substitution factor (Ramakrishnan

et al. 1999) for the subgoal at hand,

i.e., the set of free variables which exist

within the term arguments of the sub-

goal call. For example, the substitution

factor for the subgoal call p(X,1,f(Y))

is the triplet <2,X,Y>, where the �rst en-

try indicates the number of variables. The substitution factor is stored on the choice point

stack during the process of checking/inserting the call in the subgoal trie for the predi-

cate at hand. The substitution factor is then used to reduce the number of (substitution)

terms to copy into and out of the answer tries.

During evaluation, when a tabled subgoal is �rst called, it is marked as a generator call

and tagged as evaluating. At the low level engine, that corresponds to storing a generator

choice point on top of the substitution factor for the subgoal. Later, if a subgoal tagged

as evaluating is encountered again then tabling treats it as a cycle. Exactly at that part

of the tabling mechanism we intervene. Please note that at that point, we already have

the substitution factor for the repeated call also on the choice point stack. Our task is

then to unify both substitution factors in order to perform the needed uni�cation for the

coinductive strategy. Figure 4 illustrates how this mechanism works and how we use the

substitution factors to implement coinductive uni�cation. It presents the choice point and

heap stacks at the moment a cycle is found and after performing coinductive uni�cation.

This uni�cation will always succeed as YAP uses variant checking in order to identify

cycles. The next step is to follow the substitution terms for the generator choice point

and insert the resulting uni�cation as an answer in the table space for the generator call.

Finally, the execution succeeds and returns the answer.

5 Canonical Representation for Rational Terms

In this section, we address the known issue of rational terms and their canonical or

minimal representation (Colmerauer 1982). For example, the rational terms A=[1|A],

B=[1,1|B], C=[1|A] are all both uni�able and equal:

?- A=[1|A], B=[1,1|B], C=[1|A], A=B, B=C, A==B, B==C.
% SWI Prolog Result:
A = B, B = C, C = [1|_S1], % where

_S1 = [1|_S1].
% YAP Prolog Result:
Stable version: A = B = C = [1|**].
Development version: A=[1|A], B=[1,1|B], C=[1|A].

1

pair term

entry point
for A

Heap

1

pair term

entry point
for B

1

pair term

1

pair term

entry point
for C

pair term

pair term

pair term

Fig. 5. Heap representation of

rational terms A=[1|A],

B=[1,1|B], C=[1|A]

While all three lists represent the same rational term,

their underlying WAM representation in the heap may

not be the same4. Figure 5 presents YAP's representation

of each of the lists in the heap. Notice that all three

lists have a di�erent representation. This particularity of

rational terms, which does not appear in normal terms,

imposes a limitation for tabling. All these representations

appear as di�erent terms for tabling and create di�erent

subgoals and answers within the table space. As a result a

tabled predicate may table more subgoals than needed or

succeed multiple times for an answer that should succeed

only once.

In order to address the problem of the extra answers,

a possible solution would be to perform a kind of answer

subsumption. But a more elegant solution would be to

implement an algorithm that converts a rational term

to its canonical form. In this section, we present such an

algorithm. Predicate canonical_term/2 presented in Alg. 2

takes any term and converts its rational parts to their

canonical form.

4 We have veri�ed that the terms have di�erent heap representation in SWI and YAP Prolog systems.

The idea behind the algorithm is to �rst fragment the term to its cyclic subterms, con-

tinue by reconstructing each cyclic subterm (now acyclic) and, �nally, reintroduce the

cycle to the reconstructed subterms. To reconstruct each cyclic subterm as acyclic, the

algorithm copies the unique parts of the term and introduces an unbound variable instead

of the cyclic references. Then, the algorithm binds the unbound variable to the recon-

structed subterm, recreating the cycle. Take for example the rational term L=[1,2,1,2|L],

term L is being fragmented in the following subterms L0=[1|L1], L1=[2|L3], L3=[1,2|L0].

We do not need to fragment the term L3 as, at that point, our algorithm detects a cycle

and replaces term L3 with an unbound variable OpenEnd. Thus we get the following sub-

terms L0=[1|L1] and L1=[2|OpenEnd]. Binding OpenEnd=L0 results to the canonical rational

term L0=[1,2|L0].

Input: a rational term Term

Output: a rational term Canonical in canonical representation

1 canonical_term(Term, Canonical) :-
2 Term =.. InList,
3 decompose_cyclic_term(Term, InList, OutList, OpenEnd, [Term]),
4 Canonical =.. OutList,
5 Canonical = OpenEnd.
6

7 decompose_cyclic_term(_CyclicTerm, [], [], _OpenEnd, _Stack).
8 decompose_cyclic_term(CyclicTerm, [Term|Tail], [Term|NewTail], OpenEnd, Stack) :-
9 acyclic_term(Term), !,

10 decompose_cyclic_term(CyclicTerm, Tail, NewTail, OpenEnd, Stack).
11 decompose_cyclic_term(CyclicTerm,[Term|Tail],[OpenEnd|NewTail],OpenEnd,Stack) :-
12 CyclicTerm == Term, !,
13 decompose_cyclic_term(CyclicTerm, Tail, NewTail, OpenEnd, Stack).
14 decompose_cyclic_term(CyclicTerm,[Term|Tail],[Canonical|NewTail],OpenEnd,Stack) :-
15 \+ in_stack(Term, Stack), !,
16 Term =.. InList,
17 decompose_cyclic_term(Term, InList, OutList, OpenEnd2, [Term|Stack]),
18 Canonical =.. OutList,
19 ( Canonical = OpenEnd2,
20 Canonical == Term,
21 !
22 ; OpenEnd2 = OpenEnd
23 ),
24 decompose_cyclic_term(CyclicTerm, Tail, NewTail, OpenEnd, Stack).
25 decompose_cyclic_term(CyclicTerm,[_Term|Tail],[OpenEnd|NewTail],OpenEnd,Stack) :-
26 decompose_cyclic_term(CyclicTerm, Tail, NewTail, OpenEnd, Stack).

Alg. 2: Predicate canonical_term/2

The bulk of the algorithm is found at the fourth clause of decompose_cyclic_term/5.

At that part we have detected a cyclic subterm that we have to treat recursively. In

particular, lines 19�23 implement an important step. Returning to our example when

the cycle is detected, the algorithm returns the unbound variable to each fragmented

subterm. First, the subterm L1=[2|OpenEnd] appears and the algorithm needs to resolve

whether it must unify OpenEnd with L1 or whether OpenEnd must be uni�ed with a parent

subterm. In order to verify that, lines 19�23 of the algorithm unify the subterm with the

unbound variable and after attempt to unify the created rational term with the original

rational term. For our example the algorithm generates L1=[2|L1] and attempt to unify

with L=[1,2,1,2|L], as the uni�cation fails the algorithm propagates the unbound variable

to be uni�ed with the parent subterm L0=[1|L1].

The �fth clause of decompose_cyclic_term/5 is the location where a cycle is actually

found. At that point we can drop the original cyclic subterm and place an unbound

variable within the newly constructed term. The third clause of decompose_cyclic_term/5

could be omitted; it operates as a shortcut for simplifying rational terms of the form

F=f(a,f(a,F,b),b). The rest of the algorithm is pretty much straightforward, the �rst

and second clause of decompose_cyclic_term/5 are the termination condition and a copy

of the non-rational parts of the term to the new term respectively.

6 Conclusions, Related and Future Work

In this paper, we presented a novel approach that permits the insertion of rational terms

in tries and we extended tabling over goals with rational terms. Rational terms are used in

several �elds such as coinduction, de�nite clause grammars, constrain programming, lazy

evaluation, �nite state automata, etc. Our second novel contribution is a native-to-WAM

coinductive transformation. We introduced co-SLG by presenting a tabling transforma-

tion that computes the greatest �xed point that provides an e�cient elegant solution

to coinduction. Finally, we provided an ISO Prolog approach that converts any rational

term to its canonical form.

As this work spans over several di�erent �elds of logic programming, one �nds related

work from a variety of sources. In tabling, XSB Prolog (Swift and Warren 2012) has

limited support for in�nite terms within XSB's tabling mechanism. The user can set a

limit for the length of a term and de�ne an action among error, warning or abstract.

When the limit is reached the system performs the chosen action. This mechanism, while

it can ensure a cyclic safe program, is not suitable to handle rational terms as entities

neither it is suitable to be used for coinduction purposes.

In the area of coinduction, we �nd similarities in the work of (Simon et al. 2007; Gupta

et al. 2007; Moura 2013) that makes clear the importance of tabling for coinduction. (An-

cona 2013) presents an alternative approach to coinduction where tabling is not required,

still this approach may hurt readability of the coinductive predicate de�nitions. Coin-

duction for functional languages is being a topic of research for some time now (Gordon

1994) and it is been used to prove properties of lazy streams.

The problem of equivalence for �nite automata (J. E. Hopcroft and Karp 1971) is sim-

ilar with the problem of reducing a rational term to its canonical form. Recently, (Schri-

jvers and Oliveira 2012) presented an approach to check equality of rational terms in

functional programming languages. Finally, SWI Prolog (Wielemaker et al. 2012) imple-

ments a term_factorized/3 predicate that collects all subterms that are used multiple

times in a term and substitutes them by variables. While term_factorized/3 serves a

di�erent purpose, the generated subterms are in canonical form and one could use this

predicate to create the canonical form of a rational term.

As this is the �rst tabling mechanism that supports rational terms there is room for

improvements and future work. On the tabling side, we should migrate our extension to

work with more tabling strategies and options. On the side of rational terms, we intent

to further investigate transforming rational terms to their canonical form.

YAP Installation and Rational Term Support

At the time of this submission, our contributions are part of the development version

of YAP (git clone git://git.code.sf.net/p/YAP/YAP-6.3). Inside the folder YAP-6.3/ the

ICLP2014_examples.YAP �le contains the paper examples using the appropriate tabling

settings. Currently, tabling supports rational terms only when the answers are loaded by

the tries. To activate that option one can use the following YAP �ag:

yap_flag(tabling_mode, load_answers).

Furthermore, our coinductive transformation can be activated by the following directives:

:- table PREDICATE/ARITY.
:- tabling_mode(PREDICATE/ARITY, coinductive).

Technical Requirements: Our tabling extensions only require that the Prolog sys-

tem allows the creation of rational terms and that uni�cation (=/2) works between rational

terms. Our canonical_term/2 predicate also requires that the operators ==/2, =../2 work

with rational terms.

Appendix A Coinduction Examples

The following examples are recreations of the examples presented at (Moura 2013) by

using our implementation.

:- table(comember/2).
:- tabling_mode(comember/2, coinductive).

% Returns the infinite members of a list.
comember(H, L) :-

drop(H, L, L1),
comember(H, L1).

:- table(drop/3).
drop(H, [H|T], T).
drop(H, [_|T], T1) :- drop(H, T, T1).

% Queries:
?- _L=[1,2|_B], _B=[3,4,5|_B], comember(E, _L).
E = 3 ? ;
E = 4 ? ;
E = 5 ? ;
false.

?- comember(1, A).
A = [1|A] ? ;
A = [_1,1,_1|A] ? ;

A = [_1,_2,1,_1|A] ? ;
A = [_1,_2,_3,1,_1|A] ?
...

?- A=[1,2,3|A], drop(H, A, T).
A = [1,2,3|A],
H = 1,
T = [2,3,1|T] ? ;
A = [1,2,3|A],
H = 2,
T = [3,1,2|T] ? ;
A = T = [1,2,3|A],
H = 3 ? ;
false.

?- B=[1|A],A=[2,3|A], drop(H, B, T).
A = T = [2,3|A],
B = [1,2,3|B],
H = 1 ? ;
A = [2,3|A],
B = [1,2,3|B],
H = 2,
T = [3,2|T] ? ;
A = T = [2,3|A],
B = [1,2,3|B],
H = 3 ? ;
false.

?- A=[1,2,3|A], member(H, A).
A = [1,2,3|A],
H = 1 ? ;
A = [1,2,3|A],
H = 2 ? ;
A = [1,2,3|A],
H = 3 ? ;
A = [1,2,3|A],
H = 1 ? ;
A = [1,2,3|A],
H = 2 ?
...

?- B=[1|A],A=[2,3|A], member(H, B).
A = [2,3|A],
B = [1,2,3|B],
H = 1 ? ;
A = [2,3|A],
B = [1,2,3|B],
H = 2 ? ;
A = [2,3|A],
B = [1,2,3|B],
H = 3 ? ;
A = [2,3|A],
B = [1,2,3|B],
H = 2 ? ;
A = [2,3|A],
B = [1,2,3|B],
H = 3 ?
...

:- table(p/1).
:- tabling_mode(p/1, coinductive).

:- table(q/1).
:- tabling_mode(q/1, coinductive).

:- table(r/1).
:- tabling_mode(r/1, coinductive).

% Tangle example.
p([a|X]) :- q(X).
p([c|X]) :- r(X).
q([b|X]) :- p(X).
r([d|X]) :- p(X).

% Queries:
?- p(X).
X = [a,b|X] ? ;
X = [c,d|X].

?- L = [a,b,c,d|L], p(L).
L = [a,b,c,d|L].

?- L = [a,c|L], p(L).
false.

:- table(automaton/2).
:- tabling_mode(automaton/2, coinductive).

% Automaton example.
automaton(State, [Input|Inputs]) :-

trans(State, Input, NewState),
automaton(NewState, Inputs).

trans(s0, a, s1).
trans(s1, b, s2).
trans(s2, c, s3).
trans(s2, e, s0).
trans(s3, d, s0).

% Queries:
?- automaton(s0, X).
X = [a,b,c,d|X] ? ;
X = [a,b,e|X].

?- L = [a,b,c,d,a,b,e|L], automaton(s0, L).
L = [a,b,c,d,a,b,e|L].

?- L = [a,b,e,c,d|L], automaton(s0, L).
false.

:- table(sieve/2).
:- tabling_mode(sieve/2, coinductive).

:- table(filter/3).
:- tabling_mode(filter/3, coinductive).

% computes a coinductive list with all the primes in the 2..N interval
primes(N, Primes) :-

generate_infinite_list(N, List),
sieve(List, Primes).

% generate a coinductive list with a 2..N repeating patern
generate_infinite_list(N, List) :-
sequence(2, N, List, List).

sequence(Sup, Sup, [Sup| List], List) :-

!.
sequence(Inf, Sup, [Inf| List], Tail) :-
Next is Inf + 1,
sequence(Next, Sup, List, Tail).

sieve([H| T], [H| R]) :-
filter(H, T, F),
sieve(F, R).

filter(H, [K| T], L) :-
( K > H, K mod H =:= 0 ->

% throw away the multiple we found
L = T1

; % we must not throw away the integer used for filtering
% as we must return a filtered coinductive list
L = [K| T1]

),
filter(H, T, T1).

% Queries:
?- primes(20, P).
P = [2,3,5,7,11,13,17,19,2,3,5,7,11,13,17,19,2,3|P].

Appendix B Experiments

We used the following example program in order to run experiments.

:- table(path/2).
:- tabling_mode(path/2, coinductive).

% Finds infinite paths starting from node F.
path(F, [F|P]) :-

edge(F, N),
path(N, P).

edge(1, 2).
edge(1, 3).
edge(2, 4).
edge(2, 3).
edge(3, 2).

% Queries:
?- path(1, P).
P = [1,2,3|P] ? ;
P = [1,3,2|P].

?- path(2, P).
P = [2,3|P].

?- path(3, P).
P = [3,2|P].

?- path(4, P).
false.

Instead of the small graph shown above, we used a fully connected graph of di�erent

sizes, as presented next:

full_edge_size(8).

edge(X, Y) :-
posint(X),
posint(Y),
X \== Y.

posint(N) :-
posint(N, 0).

posint(_, I) :-
full_edge_size(N),
I > N, !,
fail.

posint(I, I).
posint(X, I) :-
NI is I + 1,
posint(X, NI).

And implemented the same program in Logtalk:

:- object(path).
:- public(path/2).
:- coinductive(path/2).

% :- table(path/2). % used for tabled co-SLD

path(F, [F|P]) :-
edge(F, N),
path(N, P).

full_edge_size(8).

edge(X, Y) :-
posint(X),
posint(Y),
X \== Y.

posint(N) :-
posint(N, 0).

posint(_, I) :-
full_edge_size(N),
I > N, !,
fail.

posint(I, I).
posint(X, I) :-
NI is I + 1,
posint(X, NI).

:- end_object.

We executed the query: time((path(1, _P), fail)) on graphs with size 8x8 up to 19x19.

Table B 1 presents our results; all times are in seconds. For co-SLD we used the coinduc-

tive transformation of Logtalk. For co-SLG we used our transformation. Furthermore,

we also experimented with the path/2 coinductive predicate in Logtalk being tabled with

our rational term support. This can be achieved by just tabling the co-SLD transformed

predicate of Logtalk as shown at the comment instruction of the Logtalk code.

As expected the di�erence for the path example for co-SLD and co-SLG is signi�cant.

This is easy to explain as tabling in this case decreases the complexity of the problem.

co-SLD is able to solve up to an 11x11 graph in approximately 1,700 seconds, and it

takes more than 1.5 hours which was used as our timeout to solve the 12x12 problem.

On the other hand, enumerating all the cyclic paths with co-SLG is speed e�cient but

very memory consuming. Our system exhausted the 10GB memory that was available

Table B 1. Experimental results for the query (path(1, _P), fail)

Graph Size co-SLD co-SLG Tabled co-SLD

8x8 1 0.005 4

9x9 11 0.014 *

10x10 126 0.035

11x11 1,724 0.053

12x12 > 324,000 0.126

13x13 0.287

14x14 0.674

15x15 1.5

16x16 3.5

17x17 8

18x18 17

19x19 39

20x20 *

to it trying to calculate the 20x20 graph. We also executed experiments by tabling the

co-SLD approach. These experiments, which obtained the worst results, are important in

order to point out that co-SLG is not simply achieved by tabling a co-SLD transformed

predicate. The tabled co-SLD predicate soon went out of memory as it had the task to

table both the resulting paths and the coinductive hypothesis. Unfortunately, we could

not ignore the coinductive hypothesis from being tabled due to a limitation of YAP's

mode-directed tabling (Santos and Rocha 2012).

It is well known that using tabling naively can result on a poor performance; for

example tabling append/2 predicate as shown in (Swift and Warren 2012). The same

applies for using co-SLG over co-SLD naively. There are examples that co-SLD will

perform better than co-SLG like the Eratosthenes sieve example.


